Cent. Eur. J. Math. • 8(1) • 2010 • 26-31
DOI: 10.2478/s11533-009-0073-9

VERSITA

Central European Journal of **Mathematics**

A procedure to compute prime filtration

Research Article

Asia Rauf [1*]

1 *Abdus Salam School of Mathematical Sciences, GC University, Lahore, Pakistan*

Received 6 November 2009; accepted 9 December 2009

Abstract: Let K be a field, $S = K[x_1, \dots, x_n]$ be a polynomial ring in n variables over K and $I \subset S$ be an ideal. We give a procedure to compute a prime filtration of S/I. We proceed as in the classical case by constructing an ascending chain of ideals of S starting from I and ending at S. The procedure of this paper is developed and has been implemented in the computer algebra system Singular.

MSC: 13H10, 13P10, 13C15, 13F20

Keywords: Prime filtrations • Monomial ideal • Stanley decomposition

1. Introduction

Prime filtration is a useful tool in commutative algebra. In [6, Theorem 6.4], it shows that there exists a prime filtration of a finitely generated module over a Noetherian ring. We discuss this in a special case.
Let $S = K[x_1, \dots, x_n]$ be a polynomial ring in n variables over a field K and $I \subset S$ be an ideal of S. Consider the module S/I, there exist a chain

$$\mathcal{F} : I = I_0 \subset I_1 \subset \dots \subset I_r = S$$

of ideals of S such that for each i we have $I_i/I_{i-1} \cong S/P_i$, where $P_i \subset S$ is a prime ideal for any $i \in \{1, \dots, r\}$. $\mathcal{F}$ is called a *prime filtration* of S/I and the support of $\mathcal{F}$, is the set $\mathrm{Supp}(\mathcal{F}) = \{P_1, \dots, P_r\}$. It is well known that $\mathrm{Ass}(S/I) \subset \mathrm{Supp}(\mathcal{F})$. Note that $\mathcal{F}$ gives indeed a filtration of S/I

$$(0) = I/I \subset I_1/I \subset \dots I_r/I = S/I.$$

We give a procedure to find a prime filtration of the module S/I (see Algorithm 2) and present few examples in which we compute prime filtrations using our procedure in Singular [2].
A prime filtration gives a Stanley decomposition of the module S/I, where $I \subset S$ is a monomial ideal. Let $u \in S$ be a monomial and $Z \subseteq \{x_1, \dots, x_n\}$. We denote by $uK[Z]$ the K-subspace of S/I generated by all elements uv, where v is a monomial in $K[Z]$. The multigraded K-subspace $uK[Z] \subset S/I$ is called a *Stanley space of dimension* $|Z|$, if $uK[Z]$ is a free $K[Z]$-module. A *Stanley decomposition* of S/I is a presentation of the K-vector space S/I as a finite direct sum of Stanley spaces $\mathcal{D} :\ S/I = \bigoplus_{i=1}^r u_i K[Z_i]$. Stanley decompositions and prime filtrations have been discussed in the articles [3], [5], [4], [7], [8]. A Stanley decomposition of S/I can be obtained using Algorithm 3.

* *E-mail: asia.rauf@gmail.com*

2. Computation of prime filtration

The following theorem is a special case of [6, Theorem 6.4] and we give the proof below, since we need it in our procedure.

Theorem 2.1.
Let $I \subset S$ be an ideal of the polynomial ring $S = K[x_1, \dots, x_n]$. Then S/I has a prime filtration.

Proof. Let $M = S/I$. We choose any $P_1 \in \mathrm{Ass}(M)$, then there exists a submodule M_1 of M, by definition, such that $M_1 \simeq S/P_1$. Therefore we have a chain $(0) = M_0 \subset M_1 \subset M$. If $M_1 \neq S/I$ then we may choose any $P_2 \in \mathrm{Ass}(M/M_1)$. Hence there exists a submodule $M_2 \subset M$ with $M_2/M_1 \simeq S/P_2$. So we have a chain $M_0 \subset M_1 \subset M_2 \subset M$. Continuing in the same way, this chain must stop at some M_r since M is finitely generated S-module. It implies that $M_r = M$. $\square$

We have a prime filtration of S/I of the following form:

$$I = I_0 \subset I_1 = (I, z_1) \subset I_2 = (I_1, z_2) \subset \cdots \subset I_{r-1} = (I_{r-2}, z_{r-1}) \subset I_r = (I_{r-1}, z_r) = S$$

where $z_i \in S$, $I_i/I_{i-1} = (I_{i-1}, z_i)/I_{i-1} \simeq S/P_i$ and $P_i = I_{i-1} : z_i$ for all $i \in \{1, \dots, r\}$. So to determine a prime filtration of S/I, we have to compute z_i for all $i \in \{1, \dots, r\}$. Let $I = Q_1 \cap Q_2 \cap \cdots \cap Q_m$ be a primary decomposition of I where Q_i is a primary component of I and $P_i \subset S$ is a prime ideal which is the radical of the ideal Q_i. Consider P_1 and we want to find $z \in S$ such that $I : z = P_1$. We choose z such that $z \in P_1 \setminus Q_1$ using Algorithm 1 and also $z \in \bigcap_{i=2}^{m} Q_i$.

Algorithm 1.

Input: Prime Ideal P, Ideal Q such that $\sqrt{Q} = P$
Output: Polynomial p;
- $p = 1$;
- while $(P \neq Q)$
  for($i \leq$ size(P))
    if($P[i] \notin Q$)
  $Q := Q : P[i]$;
  $p := p * P[i]$;
- return(p);

The corresponding SINGULAR code is the following:

```
proc findZ (ideal P, ideal Q, list #)
{
  int i;
  poly p = 1;
  if (size (#) > 0)
  {
      p = #[1];
  }
  if (size(reduce(P, std(Q)))== 0) // test P = Q
  {return(p);}
  for (i = 1; i <=size(P); i++)
  {
  if (reduce(P[i], std(Q))! = 0) // test P[i] ∉ Q
     {break;}
  }
  Q=quotient(Q, P[i]);
  p = p * P[i];
  return(findZ(P, Q, p));
}
```

Algorithm 2 (Prime filtration).

Input: Ideal I.
Output: A list (ideal I_{i+1}, polynomial p, ideal $I_i : p = \sqrt{L_1}$), where $L_1 \cap \ldots \cap L_s$ is a primary decomposition of I_i and $\{I_i\}$ the filtration.

- while($I \neq (1)$)
- Find polynomial p by Algorithm 1
- L=primary decomposition of $I = L_1 \cap \ldots \cap L_s$;
- $II := \bigcap_{i=2}^{s} L_i$
- $J := p$;
- $II := II \cap J$
- Choose $p \in II$ such that $p \notin L_1$
  $I := I, p$;
- return(I);

The corresponding SINGULAR code to find prime filtration is the following:

```
proc filt(ideal I,list ♯)
{
  list resu;
  if(size(♯) > 0)
  {
    resu=♯;
  }
  list L=primdecGTZ(I);// primary decomposition
  poly p =findZ(L[1][2], L[1][1]);
  ideal J = p;
  int i;
  ideal II = 1;
  for (i = 2; i <= size(L); i++)
  {
    II=intersect(II, L[i][1]);
  }
  II=intersect(II, J);
  for (i = 1; i <= size(II);i++)
  {
    if (reduce(II[i], std(L[1][1]))!=0)
    {break;}
  }
  p = II[i];
  I = I, p;
  ideal Q=std(I);
  list S = list(Q, p, L[1][2]);
  resu[size(resu)+1]=S;
  if((reduce(1, std(Q)))== 0)
    {return(resu);}
  return(filt(Q, resu));
}
```

SINGULAR Example 1.

Let $I = (x^2, y^2) \subset S = K[x, y, z]$ be an ideal.

```
LIB "primdec.lib";
ring r = 0, (x, y, z), dp;
ideal I = x2, y2;
```

filt(I);
[1] :
[1] : _[1] = $y2$ _[2] = xy _[3] = $x2$
[2] : xy
[3] : _[1] = y _[2] = x
[2] :
[1] : _[1] = y _[2] = $x2$
[2] : y
[3] : _[1] = y _[2] = x
[3] :
[1] : _[1] = y _[2] = x
[2] : x
[3] : _[1] = x _[2] = y
[4] :
[1] : _[1] = 1 _[2] = x
[2] : x
[3] : _[1] = x _[2] = y

We get the following prime filtration of S/I

$$I \subset (x^2, y^2, xy) \subset (x^2, y) \subset (x, y) \subset S.$$

SINGULAR Example 2.

LIB "primdec.lib";
ring $r = 0, (x, y, z)$, dp;
ideal $I = x3y3, z2, xyz$;
filt(I);
Hence a prime filtration of S/I is
$I \subset (z^2, xz, x^3y^3) \subset (z^2, xz, x^3y^2) \subset (z^2, xz, x^3y) \subset (z^2, xz, x^3) \subset (z, x^3) \subset (z, x^2)$
$\subset (x, z) \subset S$.

SINGULAR Example 3.

LIB "primdec.lib";
ring $r = 0$, (x, y, z, w), dp;
ideal $I = x2y - x, yzw + y2, xz + w3$;
filt(I);
We get a prime filtration of S/I
$I \subset (w^3 + xz, yzw + y^2, xzw + xy, x^2z - xw2 - xw - x, x^2y - x, xyw^2 + xyw + xy - xz, xy^2w + xy^2 - xyz + xz^2, xyz^2 - y^2w^2, xy^3 - xy^2z - xz^3 + y^2w^2, xz^5 - y^4w^2 - y^4w - y^4, y^5w^2 + y^5w + y^5 - y^4z, y^6w + y^6 - y^5z + y^4z^2, y^7 - y^6z + y^5z^2 - y^4z^3) \subset (xw + x, xy - xz, w^3 + xz, yzw + y^2, x^2z - x, xz^3 - y^2w^2, y^3w^2 + y^3w, y^4w + y^4, y^5 - y^4z) \subset (x, w^3, yzw + y^2, y^2w^2, y^3w, y^4) \subset (x, w^3, yw^2, yzw + y^2, y^2w, y^3) \subset (x, w^2, yzw + y^2, y^2w, y^3) \subset (x, w^2, yw, y^2) \subset (w, x, y^2) \subset (w, y, x) \subset S$.

3. An application

Let $I \subset S = K[x_1, \ldots, x_n]$ be a monomial ideal. Let

$$\mathcal{F} : \quad I = I_0 \subset I_1 \subset \cdots \subset I_r = S$$

be a $\mathbb{N}^n$-graded prime filtration of S/I with $I_i/I_{i-1} \cong S/P_{F_i}(-a_i)$ where $F_i \subset \{1, \ldots, n\}$, $P_{F_i} = (x_j : j \in F_i)$ and $a_i \in \mathbb{N}^n$. It was proved in [3] that this prime filtration $\mathcal{F}$ of S/I induces the Stanley decomposition

$$S/I = \bigoplus_{i=1}^{r} z_i K[Z_{F_i^c}]$$

of S/I, where $Z_{F_i^c} = \{x_j : j \notin F_i\}$ and $z_i = x^{a_i}$.

Algorithm 3 (Stanley decomposition).

Input: Monomial Ideal I.
Output: A Stanley decomposition S/I.

- m:=$(x_1, \dots, x_n)$;
- l:=filt(I);
- for ($i \leq$ size(l))
  $Q = l[i][3]$;
    $Z := \{x_j \in m : x_j \notin Q\}$;
  $S = S + l[i][2]K[Z]$
- return(S);

Here is the corresponding SINGULAR code for Stanley decomposition of S/I:

```
proc decomp(ideal I)
{
  ideal m:=maxideal(1);
  int n=size(m);
  int i;
  int j;
  list l=filt(I); // filtration of I containing all data of the Stanley decomposition
  ideal Q;
  string S;
  for (i = 1; i <=size(l); i++); // formatting the output
  {
    Q=l[i][3];
    list Z;
    for (j = 1; j <= n; j++)
    {
        if (size(reduce(var(j), std(Q)))!=0)
        {
            Z[size(Z)+1]=var(j);
        }
    }
    S = S+" + ("+string(l[i][2])+")K["+string(Z)+"]";
    kill Z;
  }
  S = S[3,size(S)];
  return(S);
}
```

SINGULAR Example 4.

Carrying on the SINGULAR Example 1. We compute a Stanley decomposition of the module S/I in SINGULAR.
decomp(I); // Stanley decomposition of S/I
$//->xyK[z] + yK[z] + xK[z] + K[z]$.

SINGULAR Example 5.

Consider the SINGULAR Example 2. Now we find a Stanley decomposition of S/I induced by the prime filtration.
decomp(I); // Stanley decomposition of S/I
$//->\ xzK[x] + x3y2K[x] + x3yK[x] + x3K[x] + zK[y] + x2K[y] + xK[y] + K[y]$.

Acknowledgements

The author would like to thank Professor Gerhard Pfister for his help and encouragement.

References

[1] Greuel G.-M., Pfister G., A Singular introduction to commutative algebra, 2nd ed., Springer-Verlag, 2007

[2] Greuel G.-M., Pfister G., Schönemann H., Singular 2.0. A computer algebra system for polynomial computations, Centre for Computer Algebra, University of Kaiserslautern, 2001, http://www.singular.uni-kl.de

[3] Herzog J., Popescu D., Finite filtrations of modules and shellable multicomplexes, Manuscripta Math., 2006, 121(3), 385–410

[4] Herzog J., Vladoiu M., Zheng X., How to compute the Stanley depth of a monomial ideal, J. Alg., 2009, 322(9), 3151–3169

[5] Jahan A.S., Prime filtrations of monomial ideals and polarizations, J. Alg., 2007, 312(2), 1011–1032

[6] Matsumura H., Commutative ring theory, Cambridge University Press, Cambridge, 1986

[7] Rauf A., Stanley decompositions, pretty clean filtrations and reductions modulo regular elements, Bull. Math. Soc. Sc. Math. Roumanie, 2007, 50(98), 347–354

[8] Rauf A., Depth and Stanley depth of multigraded modules, preprint available at http://arxiv.org/abs/0812.2080v2